

**clonevar** — Clone existing variable[Syntax](#)[Acknowledgments](#)[Menu](#)[Also see](#)[Description](#)[Remarks and examples](#)

## Syntax

```
clonevar newvar = varname [if] [in]
```

## Menu

Data > Create or change data > Other variable-creation commands > Clone existing variable

## Description

`clonevar` generates *newvar* as an exact copy of an existing variable, *varname*, with the same storage type, values, and display format as *varname*. *varname*'s variable label, value labels, notes, and characteristics will also be copied.

## Remarks and examples

[stata.com](#)

`clonevar` has various possible uses. Programmers may desire that a temporary variable appear to the user exactly like an existing variable. Interactively, you might want a slightly modified copy of an original variable, so the natural starting point is a clone of the original.

### ▶ Example 1

We have a dataset containing information on modes of travel. These data contain a variable named `mode` that identifies each observation as a specific mode of travel: air, train, bus, or car.

```
. use http://www.stata-press.com/data/r13/travel
. describe mode
```

variable name	storage type	display format	value label	variable label
mode	byte	%8.0g	travel	travel mode alternatives

```
. label list travel
travel:
      1 air
      2 train
      3 bus
      4 car
```

To create an identical variable identifying only observations that contain air or train, we could use `clonevar` with an `if` qualifier.

```
. clonevar airtrain = mode if mode == 1 | mode == 2
(420 missing values generated)
```

```
. describe mode airtrain
```

variable name	storage type	display format	value label	variable label
mode	byte	%8.0g	travel	travel mode alternatives
airtrain	byte	%8.0g	travel	travel mode alternatives

```
. list mode airtrain in 1/5
```

	mode	airtrain
1.	air	air
2.	train	train
3.	bus	.
4.	car	.
5.	air	air

The new `airtrain` variable has the same storage type, display format, value label, and variable label as `mode`. If `mode` had any characteristics or notes attached to it, they would have been applied to the new `airtrain` variable, too. The only differences in the two variables are their names and values for bus and car.



### □ Technical note

The `if` qualifier used with the `clonevar` command in example 1 referred to the values of `mode` as 1 and 2. Had we wanted to refer to the values by their associated value labels, we could have typed

```
. clonevar airtrain = mode if mode == "air":travel | mode == "train":travel
```

For more details, see [U] [13.10 Label values](#).



## Acknowledgments

`clonevar` was written by Nicholas J. Cox of the Department of Geography at Durham University, UK, and coeditor of the *Stata Journal*, who in turn thanks Michael Blasnik of M. Blasnik & Associates and Ken Higbee of StataCorp for very helpful comments on a precursor of this command.

## Also see

[D] [generate](#) — Create or change contents of variable

[D] [separate](#) — Create separate variables